Exercise 1

A right circular cone is generated by rotating a right triangle with hypotenuse 3 around one if its legs. Find the dimensions of the cone of maximum volume generated this way. What's the maximum volume?

A right triangle of hypotenuse 3. When this triangle is rotated around its horizontal leg it generates a right crcular cone of height h and radius r

```
> with(plots):
  p1:=plot([[0,0],[4,0],[0,3],[0,0]],scaling=constrained, axes=
  none,color=black):
  p2:=textplot([[-0.2,1.5,'r'],[2,-0.2,'h'],[2.1,1.7,'3']]):
  display({p1,p2});
```



Volume of the cone as a function of h= x

```
> h:=x;
  r:=sqrt(9-x^2);
  V:=(1/3)*Pi*r^2*h;
```

$$h := x$$
$$r := \sqrt{9 - x^2}$$
$$V := \frac{\pi \left(9 - x^2\right) x}{3} \qquad \textbf{(1)}$$

The domain of is the interval [0,3]. Plot of the function V(x)

```
> plot(V,x=0..3);
```



We observe that V reaches an absolute maximum in the interval [0,3]. To identify the maximum we look for the critical points of V in the interval (0,3)

```
> diff(V,x);
  s:=solve(diff(V,x),x);
```

$$-\frac{2\pi x^2}{3} + \frac{\pi(9 - x^2)}{3} \quad (2)$$

$$s := \sqrt{3}, -\sqrt{3}$$

Only the positive value is in the interval (0,3).

```
> x0:=s[1];
```

$$x0 := \sqrt{3} \quad (3)$$

Dimensions and Volume of the cone of maximum volume

```
> h0:=subs(x=x0,h);
  r0:=subs(x=x0,r);
  Vmax:=subs(x=x0,V);
```

$$h0 := \sqrt{3} \quad (4)$$

$$r0 := \sqrt{6}$$

$$Vmax := 2\pi\sqrt{3}$$

Exercise 2

The base of a solid is the region bounded by the curve y=exp(-x) and the lines x=0,y=0 and x=ln 2. Find the volume of the solid if its cross sections parallel to the yz-plane are circles with a diameter running between the curve and the x-axis.

Plot of the base and the diameter of the circular cross section

```
> with(plots):
  p1:=plot(exp(-x),x=0..ln(2),tickmarks=[0,0]):
  p2:=plot([ln(2)/2,t,t=0..exp(-ln(2)/2)]):
  p3:=plot([ln(2),t,t=0..exp(-ln(2))]):
  p4:=textplot([[ln(2)/2,-0.1,'x'],[ln(2)/2,exp(-ln(2)/2)+0.1,'exp
  (-x)'],[ln(2),-0.1,'ln2']]):
  display({p1,p2,p3,p4});
```



The cross sections are circles of diameter D=exp(-x). The area of the cross section at x is

```
> r:=exp(-x)/2;
  A:=simplify(Pi*r^2);
```

$$r := \frac{1}{2}\,e^{-x}$$

$$A := \frac{1}{4}\,\pi\,e^{-2x} \quad \textbf{(5)}$$

The volume of the solid is given by the integral of this function A(x) from x=0 to x=ln2

```
> V:=Int(A,x=0..ln(2));
  value(V);
```

(6)

$$V := \int_0^{\ln(2)} \frac{1}{4} \pi e^{-2x} \, dx \quad (6)$$

$$\frac{3\pi}{32}$$

Exercise 3

a) Study convergence of the series of general term a(n)=3^n/n^2, with n from 1 to infinity and find the sum if convergent.

b) Repeat with the series of general term cos(nPi)/4^n

a) General term of the series

```
> a:=n->3^n/n^2;
```

$$a := n \to \frac{3^n}{n^2} \quad (7)$$

Apply the ratio test

```
> L:=Limit(a(n+1)/a(n),n=infinity);
```

$$L := \lim_{n \to \infty} \frac{3^{n+1} n^2}{(n+1)^2 3^n} \quad (8)$$

```
> L:=Limit(simplify(a(n+1)/a(n)),n=infinity);
```

$$L := \lim_{n \to \infty} \frac{3 n^2}{(n+1)^2} \quad (9)$$

```
> L:=limit(simplify(a(n+1)/a(n)),n=infinity);
```

$$L := 3 \quad (10)$$

Since L>1 the series diverges

b) The values of cos(n*Pi) alternate between -1 and +1

```
> seq(cos(n*Pi),n=1..5);
```

$$-1, 1, -1, 1, -1 \quad (11)$$

General term of the series

```
> a:=n->(-1)^n/4^n;
```

$$a := n \to \frac{(-1)^n}{4^n} \quad (12)$$

Series

```
> Sum(a(n),n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{(-1)^n}{4^n} \quad (13)$$

This is a geometric series of ratio r=-1/4 without the initial term r^0=1. Since abs(r)<1 the series converges and the sum from N=1 to infinity is given by

```
> r:=-1/4;
  S:=1/(1-r)-1;
```

$$r := -\frac{1}{4} \quad (14)$$

$$S := -\frac{1}{5}$$

Checking the sum

```
> sum(a(n),n=1..infinity);
```

$$-\frac{1}{5} \quad \textbf{(15)}$$